query: embed.html#SELECT distinct ?const ?constLabel ?parent ?parentLabel ?start ?layer ?coords ?colour ?rgb
  with {
    SELECT ?colour ?rgb WHERE {
      VALUES (?colour ?rgb) {
        (0.0 '0000FF')(1.0 '002BFF')(2.0 '0055FF')(3.0 '007FFF')(4.0 '00AAFF')(5.0 '00D5FF')
        (6.0 '00FFFF')(7.0 '00FFD5')(8.0 '00FFAA')(9.0 '00FF80')(10.0 '00FF55')(11.0 '00FF2B')
        (12.0 '00FF00')(13.0 '2BFF00')(14.0 '55FF00')(15.0 '80FF00')(16.0 'AAFF00')(17.0 'D4FF00')
        (18.0 'FFFF00')(19.0 'FFD500')(20.0 'FFAA00')(21.0 'FF8000')(22.0 'FF5500')(23.0 'FF2A00')
        (24.0 'FF0000')      
      }
    }
  } AS %rgb_colours
  
  with {
    SELECT distinct ?const  ?parent  ?start ?layer ?coords ?colour WHERE {
      ?item p:P39 ?positionStatement . ?positionStatement ps:P39 ?term . ?term wdt:P279 wd:Q18018860 .
      ?positionStatement pq:P768 ?const .
      ?positionStatement prov:wasDerivedFrom ?ref . ?ref pr:P1614 ?refhop .
      optional { ?const wdt:P131 ?parent .
        optional { ?parent wdt:P625 ?coords } }
      ?const wdt:P571 ?start .
      FILTER (year(?start) > 1500) .
      BIND (floor((year(?start) - 1500)/10) AS ?colour) .
      BIND (CONCAT(str(10 * floor(year(?start)/10)),'s') AS ?layer) .
    }
  } AS %consts 
  
  WHERE {
      INCLUDE %consts .
      INCLUDE %rgb_colours .
      SERVICE wikibase:label { bd:serviceParam wikibase:language 'en' }
  } order by ?start
  #defaultView:Map